SUPPLEMENTARY MATERIAL

TABLE S1. ^{1}H and ^{13}C chemical shifts (±0.002ppm) for strychnine in deuterated chloroform at 25°C. The chemical shifts are referenced relative to internal chloroform at 7.284ppm (^{1}H) and 77.070ppm (^{13}C).

Strychnine				
	pro-R* / *pro-S	**δ /ppm**		**δ /ppm**
H13		1.280	C15	26.878
H15a	*pro-R*	1.460	C14	31.619
H17b	*pro-R*	1.893	C11	42.508
H17a	*pro-S*	1.897	C17	42.898
H15b	*pro-S*	2.364	C13	48.247
H11b	*pro-R*	2.679	C18	50.363
H20b	*pro-S*	2.729	C7	51.957
H18b	*pro-R*	2.875	C20	52.689
H11a	*pro-S*	3.141	C8	60.127
H14		3.148	C16	60.208
H18a	*pro-S*	3.196	C23	64.625
H20a	*pro-S*	3.708	C12	77.618
H8		3.865	C4	116.229
H16		3.944	C1	122.281
H23a	*pro-S*	4.071	C2	124.212
H23b	*pro-R*	4.156	C22	127.191
H12		4.293	C3	128.548
H22		5.903	C6	132.812
H2		7.102	C21	140.615
H1		7.168	C5	142.218
H3		7.261	C10	169.312
H4		8.102		

TABLE S2. ^{1}H and ^{13}C chemical shifts (±0.002ppm) for camphor in deuterated chloroform at 25°C. The chemical shifts are referenced relative to internal chloroform at 7.284ppm (^{1}H) and 77.044ppm (^{13}C).

Camphor				
	pro-R* / *pro-S	**δ /ppm**		**δ /ppm**
H10		0.834	C6	9.249
H6		0.913	C9	19.144
H9		0.960	C10	19.779
H3b	*pro-S*	1.341	C3	27.048
H4b	*pro-S*	1.405	C4	29.914
H4a	*pro-R*	1.683	C2	43.045
H1b	*pro-S*	1.845	C1	43.299
H3a	*pro-R*	1.952	C8	46.789
H2		2.090	C5	57.697
H1a	*pro-R*	2.351	C7	219.684

TABLE S3. Summary of all calculated/experimental/literature ${}^{n}J_{CH}$ values for which a value could be measured in coupled ${}^{13}C$ spectra. A) Absolute DFT + 6% correction B) Coupled ${}^{13}C$ (fully or selectively coupled) C) IPAP accordion HSQMBC, J_{LR}=3-8Hz D) IPAP PIP-HSQMBC, J_{LR}=6Hz E) IPAP analysis of refocused HSQMBC, J_{LR}=6Hz F) IPAP HSQMBC-COSY, J_{LR}=6Hz G) IPAP HMBC, J_{LR}=6Hz H) IPAP HMBC-COSY, J_{LR}=6Hz I) EXSIDE, J_{LR}=6Hz, N=30, tilted with equation 3 J) *J*-HMBC, J_{LR}=1Hz, κ=56 (camphor) or 79 (strychnine) K) Lineshape analysis of PIP-HSQMBC, J_{LR}=8Hz L) Literature.

		A DFT	**B Coupled ${}^{13}C$**	**C IPAP ac. HSQMBC**	**D IPAP PIP-HSQMBC**	**E IPAP Refocussed HSQMBC**	**F IPAP HSQMBC-COSY**	**G IPAP HMBC**	**H IPAP HMBC-COSY**	**I EXSIDE (tilted)**	**J *J*-HMBC**	**K PIP-HSQMBC**	**L Literature [1,2,3]**
MAD / Hz		0.38		0.33	0.22	0.33	0.27	0.35	0.34	0.14	0.28	0.18	0.62
SD / Hz		0.49		0.45	0.34	0.52	0.42	0.40	0.44	0.19	0.44	0.14	0.84
		$\|{}^{n}J_{HC}\|$ /Hz	**$\|{}^{n}J_{HC}\|$ (Error†) /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ /Hz**	**${}^{n}J_{HC}$ (SD) /Hz**
Camphor													
C1	H2	0.8	0.5 (0.3)										
C1	H3a	9.0	9.3 (0.2)								9.1		
C1	H3b	5.6	5.2 (0.2)										
C1	H4a	0.8	0.4 (0.4)										
C1	H4b	0.2	0.7 (0.3)										
C2	H1a	2.4	2.7 (**0.6**)							2.4	2.5		
C2	H1b	2.4	2.9 (**0.8**)							2.7	2.4		
C2	H3a	3.2	2.0 (**0.7**)								2.8		
C2	H3b	3.0	0.3 (**2.0**)‡										
C2	H4a	1.3	1.6 (**0.6**)										
C2	H4b	1.6	3.2 (**0.8**)								1.1		
C4	H2	6.6	6.9 (0.3)	6.8	6.8	6.9		6.8		7.0			
C4	H3a	3.9	4.0 (0.4)			3.6		3.9		3.6	3.6		
C4	H3b	2.2	2.3 (0.3)	3.1	3.1	3.3		3.2					
C4	H6	4.3	4.3 (0.5)	4.4	4.4	4.4		4.4		4.5	4.5		
C6	H2	0.5	0.6 (0.2)	1.9	1.6	1.5		1.8					

C6	H4a	3.2	3.0 (0.2)	3.1	2.9	3.0		3.2		3.1	2.7		
C7	H1a	6.8	6.4 (0.2)	6.4	6.3	6.4		6.4		6.5	6.4		
C7	H1b	4.9	4.3 (1.0)	4.7	4.6	4.7		4.7		4.7	4.6		
C7	H2	8.5	8.5 (1.3)	8.4	8.6	8.7		8.4		8.6	1.0 §		
C7	H3a	0.9	0.3 (1.4)					0.9			1.0		
C7	H3b	0.3	0.8 (2.0)‡					0.2					
C7	H4a	7.3	8.2 (0.9)	7.7	7.8	7.6		8.0		7.8	7.4		
C7	H4b	5.1	4.1 (1.6)	4.5		4.5		4.6			4.4		
C7	H6	3.9	4.2 (1.5)	4.0	4.0	4.0		4.1		4.1	4.0		
C8	H1a	0.6	0.6 (**0.9**)	1.1	0.6	1.6		0.8			1.4		
C8	H1b	8.6	8.5 (**0.4**)	8.6	8.6	8.5		8.5		8.6	8.8		
C8	H3b	6.3	6.1 (**0.7**)	6.5	6.4	6.5		6.6			6.4		
C8	H4b	6.0	6.5 (**0.6**)	6.3	6.4	6.3		6.3			6.4		
C9	H10	4.8	4.9 (0.3)	4.7	4.7	4.8		4.5		5.0	4.9		
C9	H1b	0.9	1.0 (0.5)	1.4				1.6			1.0		
C9	H2	1.3	1.2 (0.7)	2.3	1.7	1.9		1.9			1.5		
C10	H1b	0.4	0.6 (0.6)	1.3				1.4					
C10	H2	0.9	0.9 (0.4)	1.9	1.6	1.7		1.7					
C10	H9	4.8	5.0 (1.1)	4.9	4.9	4.9		4.9		5.1	5.0		
$^nJ_{CH}$ Total		35	35	21	18	20		24		14	23		
$^nJ_{CH}$<1Hz		10	10	0	1	0		3		0	0		
Strychnine													
C1	H2	0.7	0.9 (0.6)	1.4				1.4			0.9		
C1	H3	8.5	8.9 (0.4)	8.8	9.1			9.0			9.2		
C1	H4	1.4	1.1 (1.1)	1.3		0.6	0.6	1.8	1.6				
C1	H8	0.3	1.3 (1.3)	1.3	1.6		0.9						
C2	H1	0.5	0.3 (0.9)										

C2	H3	0.4	1.3 (0.9)								1.8		
C2	H4	7.4	7.6 (0.4)					7.6		7.6	7.8		7.6 (0.3)
C4	H1	1.4	1.1 (0.2)			1.5	1.7	1.8	1.7		1.1		
C4	H2	7.6	8.0 (0.3)	7.7	7.9	7.7		7.8			7.9		
C4	H3	1.2	2.1 (0.2)	2.3	2.3			2.2	2.4		2.0		
C4	H8	0.1	0.4 (0.5)										
C5	H1	7.9	8.0 (0.9)	8.0	8.0	8.3	8.2	7.9			8.1		
C5	H2	1.6	1.3 (0.5)	1.9		1.4	1.3	1.6	1.5		0.8		
C5	H3	9.7	9.7 (1.0)	10.0	10.1			9.8			10.0		
C5	H4	3.4	2.2 (1.1)	2.2			2.3	2.9		2.2	2.0		1.1 (0.4)
C5	H8	3.0	3.2 (1.0)	3.2	3.2	3.3	3.2	3.3	3.3	3.1	3.0	3.4	2.9 (0.4)
C5	H11b	0.7	0.6 (0.6)				0.1	1.3			0.8		
C5	H12	0.1	1.0 (1.0)						0.4				
C5	H13	0.1	0.4 (0.6)			0.3	0.4		0.1				
C10	H8	0.1	0.6 (0.7)										
C10	H11a	6.6	6.4 (0.5)	6.4	6.4	6.4	6.4		6.4	6.4	6.5	6.5	6.3 (0.3)
C10	H11b	8.4	7.9 (0.5)	8.0	7.9	7.9	8.0	7.9	8.0	7.9	8.0		7.6 (0.4)
C10	H12	1.5	2.0 (0.5)	2.1	2.1	1.9	2.1	2.6	2.1		1.8	2.4	3.8 (2.9)
C11	H12	2.0	1.9 (0.3)	2.2	1.8						1.7		
C11	H13	0.7	0.6 (1.1)										
C15	H13	3.8	3.2 (1.8)	3.4	3.5			3.6		3.4	3.3		3.7 (0.5)
C15	H14	3.9	4.0 (1.7)			1.9	2.6		4.9	3.8			
C15	H16	2.0	1.2 (1.5)	2.3				2.6					
C15	H20b	1.4	1.9 (0.9)	2.0	1.7	1.8	1.9		1.7		1.4		
C17	H8	6.3	5.9 (0.7)	5.9	5.9	5.9			6.0	5.8	5.7		5.7 (0.3)
C17	H13	0.1	0.7 (1.3)			0.2	0.1		0.1				
C17	H18a	1.7	1.6 (0.7)	2.6					2.6				3.4 (0.5)
C17	H18b	4.7	4.6 (0.5)	5.1	4.7	4.7	4.7			4.8	5.0		5.1 (0.5)

C17	H20b	0.1	1.0 (1.0)										
C22	H14	3.8	3.5 (1.6)							3.9			
C22	H15a	0.5	0.9 (0.9)										
C22	H20a	4.8	5.4 (0.6)	5.5	5.4	5.5	5.4	5.4	5.8	5.4	5.3	5.5	5.7 (0.5)
C22	H20b	4.8	4.7 (1.1)	4.7	4.6	4.6	4.7	4.8	4.7	4.7	4.7	4.7	4.4 (0.2)
C22	H23a	3.4	3.2 (1.1)	3.4	3.2	3.2		3.3	3.3		3.4		
C22	H23b	4.1	4.0 (1.1)	3.8		3.7	3.4	3.8	3.7		3.7		
C23	H12	3.1	2.7 (0.4)	2.6	2.6	3.0	2.7	2.9		2.7	2.8		3.5 (2.2)
C23	H20a	1.5	1.2 (1.2)	3.0	2.3						1.1		
C23	H20b	0.3	0.9 (1.6)										
C23	H22	6.4	7.0 (0.6)	7.2				6.9	7.5	7.0		7.3	6.4 (0.5)
$^{n}J_{CH}$ Total		44	44	29	20	20	20	23	20	14	27	6	14
$^{n}J_{CH}$<1Hz		14	11	0	0	3	5	0	3	0	3	0	0

† HESSE error in fitting of $^{n}J_{CH}$ reported from fitting fully coupled or selectively decoupled (**bold**) ^{13}C spectrum.
‡ Not included in MAD/SD calculation due to large (≥2.0Hz) HESSE error.
§ MAD/SD calculated including this value are 0.42/1.14Hz



FIGURE S4. EXSIDE for Camphor A) Raw J_{LR}=6Hz B) Matched C) Camphor correction for J_{LR}=6Hz [corrected by $^nJ_{CH}$(scaled) = ($^nJ_{CH}$(EXSIDE, camphor J_{LR} = 6 Hz) + 1.619) / 1.263)]

$$^nJ_{CH}(\text{scaled}) = (^nJ_{CH}(J_{LR}) - \text{intercept}(J_{LR})) / \text{gradient}(J_{LR})) \qquad \text{(Equation 3)}$$

TABLE S5. Gradients and intercepts for use in equation 3 for tilting of EXSIDE data with different J_{LR}.

J_{LR} /Hz	10	8	6	4	2
Gradient(J_{LR})	1.158	1.194	1.256	1.186	0.951
Intercept(J_{LR})	-1.542	-1.593	-1.580	-0.787	0.373



FIGURE S6. Matched EXSIDE for strychnine and camphor compared to coupled ^{13}C, MAD/SD 0.19/0.24Hz.



A
y = x
y = 1.158x - 1.542
J_{LR}=10Hz
$^nJ_{CH}$ Raw EXSIDE / Hz
$|^nJ_{CH}|$ Coupled ^{13}C / Hz
B
y = x
J_{LR}=10Hz
$^nJ_{CH}$ Tilted EXSIDE / Hz
$|^nJ_{CH}|$ Coupled ^{13}C / Hz
C
y = x
y = 1.194x - 1.593
J_{LR}=8Hz
$^nJ_{CH}$ Raw EXSIDE / Hz
$|^nJ_{CH}|$ Coupled ^{13}C / Hz
D
y = x
J_{LR}=8Hz
$^nJ_{CH}$ Tilted EXSIDE / Hz
$|^nJ_{CH}|$ Coupled ^{13}C / Hz
E
y = x
y = 1.256x - 1.580
J_{LR}=6Hz
$^nJ_{CH}$ Raw EXSIDE / Hz
$|^nJ_{CH}|$ Coupled ^{13}C / Hz
F
y = x
J_{LR}=6Hz
$^nJ_{CH}$ Tilted EXSIDE / Hz
$|^nJ_{CH}|$ Coupled ^{13}C / Hz



FIGURE S7. Raw (blue) and tilted (green) EXSIDE for strychnine and camphor, tilted with equation 3.



FIGURE S8. Raw (blue) and tilted (green) EXSIDE for strychnine with a scaling factor (N) of 15, tilted with equation 3 for J_{LR}=8Hz.



FIGURE S9. Basis of DFT correction factor by comparison to coupled ^{13}C A) uncorrected DFT, MAD/SD: 0.43/0.49Hz B) DFT with a 6% correction, MAD/SD: 0.38/0.49Hz.

TABLE S10. Cartesian coordinates for geometry optimised (DFT mPW1PW91/6-311g(d,p) IEFPCM(Chloroform)) structure of camphor.

Atom Type	Coordinates /Å		
	x	y	z
C	-0.859113	1.246945	1.064808
C	0.60069	1.256985	0.596464
C	0.58901	1.770202	-0.852523
C	-0.067271	0.598084	-1.62376
C	-0.291295	-0.501578	-0.544223
C	-0.542439	-1.887506	-1.084096
C	-1.433622	0.071131	0.2826
C	0.910804	-0.256273	0.420035
C	2.275462	-0.531861	-0.20628
C	0.819292	-1.067798	1.713862
H	0.021228	2.698424	-0.941768
H	1.596763	1.974828	-1.216391
H	-1.004978	0.877707	-2.108865
H	0.586719	0.204497	-2.404539
H	-1.405916	2.164887	0.837318
H	-0.973715	1.060749	2.135011
H	1.281043	1.790901	1.261405
H	1.007668	-2.125148	1.511867
H	1.576541	-0.731203	2.427055
H	-0.152708	-1.002633	2.206015
H	3.068094	-0.201255	0.471007
H	2.412047	-1.603826	-0.369803
H	2.427376	-0.030186	-1.161827
H	-1.459971	-1.903164	-1.676249
H	0.280531	-2.216097	-1.722891
H	-0.663678	-2.6141	-0.277602
O	-2.575768	-0.316796	0.303617

TABLE S11. Cartesian coordinates for geometry optimised (DFT mPW1PW91/6-311g(d,p) IEFPCM(Chloroform)) structure of strychnine.

Atom Type	Coordinates /Å x	y	z
C	2.342741	0.438577	0.169109
C	3.651101	0.875689	-0.003272
C	4.611301	-0.072003	-0.350256
C	4.279870	-1.412626	-0.512476
C	2.965530	-1.835790	-0.320263
C	1.997114	-0.906401	0.019608
C	0.547175	-1.117499	0.362818
C	-0.002709	0.334617	0.410041
C	-0.266232	-2.066879	-0.548852
C	-1.094173	-2.235162	1.699038
C	0.366591	-1.815202	1.723354
C	-0.816951	-1.402860	-1.801806
C	-1.729466	-0.247480	-1.381719
C	-0.809549	0.826699	-0.791995
C	-2.700442	-2.240822	-0.146570
C	-2.798976	-0.767050	-0.435862
C	-3.693739	0.032444	0.143122
C	-3.687027	1.522227	-0.038134
C	-1.410937	2.184736	-0.415514
C	-0.357575	3.051279	0.305734
C	1.070697	2.545316	0.312218
N	1.197811	1.192093	0.476776
N	-1.355643	-2.600918	0.303180
O	2.023962	3.297435	0.204358
O	-2.506325	2.122315	0.480709
H	3.899224	1.918342	0.120300
H	5.636294	0.248726	-0.495160
H	5.044523	-2.130892	-0.782176
H	2.708769	-2.883184	-0.432261
H	-0.602008	0.476487	1.313292
H	0.394181	-2.892515	-0.838894
H	-1.734309	-1.398610	2.017085
H	-1.296452	-3.074790	2.369850
H	1.020771	-2.690170	1.762928
H	0.616763	-1.167611	2.565450
H	0.008537	-1.034130	-2.415752
H	-1.354123	-2.136297	-2.406870
H	-2.220151	0.177103	-2.266332
H	-0.082901	1.048831	-1.584748
H	-2.949513	-2.826547	-1.037655
H	-3.413761	-2.534616	0.625290
H	-4.432181	-0.391446	0.817810
H	-4.514373	1.976844	0.507272
H	-3.802258	1.794498	-1.097361
H	-1.739818	2.686128	-1.335697
H	-0.681035	3.120661	1.349222
H	-0.328511	4.066099	-0.085762

Table S12. Summary of total experiment time and experimental details.

Experiment	Total time*	Number of scans	t1 increments	F1 spectral width /Hz	t2 increments	F2 spectral width /Hz
Camphor						
Fully coupled ^{13}C	1h47m	2048	65536	29761.9	N/A	N/A
Selectively decoupled ^{13}C	52m	1024	65536	31250.0	N/A	N/A
EXSIDE (N=30)**	2h15m	2	963	28901.7	1500	10000.0
IPAP Accordion HSQMBC	2h45m	4	716	16339.9	8192	4000.0
IPAP HMBC	6h28m	8	1280	28901.7	8192	4000.0
IPAP PIP-HSQMBC	6h20m	8	1280	28901.7	8192	4000.0
PIP-HSQMBC	3h10m	8	640	28901.7	8192	4000.0
IPAP refocussed HSQMBC	6h18m	8	1280	28901.7	8192	4000.0
J-HMBC (κ=56)	3h29m	8	512	28933.0	8192	4000.0
Strychnine						
Fully coupled ^{13}C	2h5m	3072	71680	25252.5	N/A	N/A
Selectively decoupled ^{13}C	59m	1172	66560	31250.0	N/A	N/A
EXSIDE (N=30)	3h44m	4	1280	19531.3	512	2500.0
EXSIDE (N=15, J_{LR}=8)**	3h18m	2	1592	23873.5	1500	10000.0
IPAP Accordion HSQMBC	2h32m	4	1024	19531.3	8192	4401.4
IPAP HMBC	3h41m	12	512	19531.3	12288	6321.1
IPAP HMBC COSY	3h41m	12	512	19531.3	12288	6321.1
IPAP PIP-HSQMBC	7h23m	12	1024	19531.3	12288	6321.1
PIP-HSQMBC	3h42m	12	512	19531.3	12288	6321.1
IPAP refocussed HSQMBC	3h41m	12	512	19531.3	12288	6321.1
HOBS HSQMBC	1h50m	8	256	19531.3	10240	6321.1
J-HMBC (κ=79)	1h35m	8	256	20127.5	8192	5498.5

* For EXSIDE, IPAP HSQMBC and IPAP HMBC experiments with J_{LR}=6Hz, *J*-HMBC with J_{LR}=1Hz and IPAP Accordion HSQMBC with J_{LR}=3-8Hz unless otherwise indicated.

** Recorded on Varian VNMRS 500MHz Direct Drive Spectrometer with Agilent OneNMR probe while all other experiments were recorded Bruker AVANCE III HD 500MHz NMR Spectrometer with 5mm DCH ^{13}C-^{1}H/D Cryo Probe. Therefore the increments are half what they would appear on a Bruker spectrometer.



FIGURE S13. H2-C7 for Camphor, the ^{1}H spectrum for H2 is shown along f2 and the dashed line indicates the position of the f1 trace. A) J-HMBC, $^{3}J_{H2\text{-}C7}$ = 1.0Hz, J_{LR}=1Hz, κ=56 B) EXSIDE, $^{3}J_{H2\text{-}C7}$ = 9.2Hz (8.4Hz after tilting), J_{LR}=6Hz, N=30.

Pulse Sequences

The following sequences from the Bruker pulse program library were used:

1) EXSIDE

```
;hsqcetgplrjcsp
;avance-version (13/01/23)
;EXSIDE
;2D H-1/X correlation via double inept transfer
;for determination of XH long range coupling constants
;   via splitting in F1
;phase sensitive using Echo/Antiecho-TPPI gradient selection
;using selective pulses for H-1
;with decoupling during acquisition
;
;V.V. Krishnamurthy, J. Magn. Reson., Series A 121, 33-41 (1996)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"d2=1s/(cnst2*2)"
"d11=30m"

"d0=3u"
"d20=d6/4-p16-d16-4u-larger(p12,p14)/2"

"in0=inf1/2"

"in20=0.5*in0*cnst16"

"DELTA=p16+d16+p2+d0*2-4u"
"DELTA1=d6/4-p16*2-d16*2-4u-larger(p12,p14)/2"
"DELTA2=d6/4-p16-d16-4u-larger(p12,p14)/2"
"DELTA3=d2-larger(p2,p14)/2"

1 ze
  d11 pl12:f2
2 d1 do:f2
3 (p1*1.5 ph1)
  DELTA3
  (center (p2 ph8) (p14:sp3 ph1):f2 )
  DELTA3 UNBLKGRAD
  (p1*0.5 ph1)
  4u
  p16:gp3
  d16

  (p1 ph1)

  d20
  4u
  p16:gp4
  d16
  (center (p12:sp2 ph1) (p14:sp3 ph7):f2 )
  4u
  p16:gp4
  d16
  d20
```

```
  d20
  4u
  p16:gp5
  d16
  (center (p12:sp2 ph3) (p14:sp3 ph7):f2 )
  4u
  p16:gp5
  d16 pl1:f1
  d20

  (p1 ph2)

  4u
  p16:gp6
  d16 pl2:f2

  (p3 ph4):f2
  d0
  p2 ph6
  d0
  p16:gp1*EA
  d16
  (p24:sp7 ph5):f2
  4u
  DELTA pl2:f2
  (ralign (p1 ph1) (p3 ph5):f2 )

  4u
  p16:gp2
  d16
  DELTA1
  p16:gp4
  d16
  (center (p12:sp2 ph1) (p14:sp3 ph1):f2 )
  4u
  p16:gp4
  d16
  DELTA2

  DELTA2
  4u
  p16:gp5
  d16
  (center (p12:sp2 ph3) (p14:sp3 ph1):f2 )
  4u
  p16:gp5
  d16 pl1:f1 pl12:f2
  DELTA2 BLKGRAD
  go=2 ph31 cpd2:f2
  d1 do:f2 mc #0 to 2
    F1EA(calgrad(EA), caldel(d0, +in0) & caldel(d20, +in20) & calph(ph4, +180) & calph(ph7, +180) &
calph(ph31, +180))
exit

ph1=0
ph2=1
ph3=2
ph4=0 2
ph5=0 0 0 0 2 2 2 2
ph6=0 0 2 2
ph7=0
ph8=2
ph31=0 2 0 2 2 0 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
```

```
;sp2: f1 channel - shaped pulse 180 degree
;sp3: f2 channel - shaped pulse (180degree inversion)
;spnam3: Crp60,0.5,20.1
;sp7: f2 channel - shaped pulse (180degree refocussing)
;spnam7: Crp60comp.4
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p12: f1 channel - 180 degree shaped pulse (selective)
;p14: f2 channel - 180 degree shaped pulse for inversion
;     = 500usec for Crp60,0.5,20.1
;p16: homospoil/gradient pulse
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;d0 : incremented delay (2D)                    [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2J)XH
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d11: delay for disk I/O                        [30 msec]
;d16: delay for homospoil/gradient recovery
;d20: incremented delay (J-evolution)
;cnst2: = J(XH)
;cnst16: = J(scale) factor
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;in20: = in0 * J(scale)
;ns: 4 * n
;ds: >= 16
;td1: number of experiments
;FnMODE: echo-antiecho
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;use gradient ratio:        gp 1 : gp 2 : gp 3 : gp 4 : gp 5 : gp 6
;                             80 : 20.1 :   20 :    7 :   11 :   50   for C-13

;for z-only gradients:
;gpz1: 80%
;gpz2: 20.1% for C-13
;gpz3: 20%
;gpz4: 7%
;gpz5: 11%
;gpz6: 50%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100
;gpnam6: SMSQ10.100

;$Id: hsqcetgplrjcsp,v 1.7 2013/02/05 10:11:25 ber Exp $
```

2) Coupled ^{13}C (zggd30)

```
;zggd30
;avance-version (12/01/11)
;1D sequence with gated decoupling
;using 30 degree flip angle
;
;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>

"d11=30m"

"acqt0=-p1*0.66/3.1416"

1 ze
  d11 pl12:f2
2 30m
  4u cpd2:f2
  d1
  p1*0.33 ph1
  go=2 ph31 do:f2
  30m mc #0 to 2 F0(zd)
exit

ph1=0 2 2 0 1 3 3 1
ph31=0 2 2 0 1 3 3 1

;pl1 : f1 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                          [30 msec]
;ns: 1 * n, total number of scans: NS * TD0
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;$Id: zggd30,v 1.11 2012/01/31 17:49:32 ber Exp $
```

3) Selectively ^{1}H decoupled ^{13}C (zgig)

```
;zgig
;avance-version (12/01/11)
;1D sequence with inverse gated decoupling
;
;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>

"d11=30m"

"acqt0=-p1*2/3.1416"

1 ze
  d11 pl12:f2
2 30m do:f2
  d1
  p1 ph1
  go=2 ph31 cpd2:f2
  30m do:f2 mc #0 to 2 F0(zd)
exit

ph1=0 2 2 0 1 3 3 1
ph31=0 2 2 0 1 3 3 1

;pl1 : f1 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  high power pulse
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                         [30 msec]
;ns: 1 * n, total number of scans: NS * TD0
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;$Id: zgig,v 1.11 2012/01/31 17:49:32 ber Exp $
```

4) *J*-HMBC (hmbcetgpjcl2nd)

```
;hmbcetgpjcl2nd
;avance-version (12/01/11)
;J-HMBC
;2D H-1/X correlation via heteronuclear zero and double quantum
;   coherence
;for determination of XH long range coupling constants
;   via splitting in F1
;phase sensitive using Echo/Antiecho gradient selection
;with two-fold low-pass J-filter to suppress one-bond correlations
;
;A. Meissner & O.W. Soerensen, Magn. Reson. Chem. 39, 49-52 (2001)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"cnst30=(1-sfo2/sfo1)/(1+sfo2/sfo1)"

define list<gradient> EA1 = { 1.000 -cnst30}
define list<gradient> EA2 = { -cnst30 1.000}

"p2=p1*2"
"d6=1s/(cnst13*2)"

"in0=inf1/2"

"SCALEF=(d6/2-p14/2-3u)*2/(in0*td1)"

"in20=in0"
"in28=in0*(1+SCALEF)"
"in29=in0*SCALEF"

"d0=3u"
"d20=(d6-1s/(cnst6*2)-1s/(cnst7*2))/2+(in0*td1)-p3-p16*2-d16*2"
"d28=(d6-p14)/2+(in0*td1)"
"d29=(1s/(cnst6*2)+1s/(cnst7*2))/2+p3+p16+d16-p14/2+p24+p2+d0*2"

"DELTA1=1s/(cnst6*2)-p16-d16"
"DELTA2=1s/(cnst7*2)-p16-d16"
"DELTA3=p2+d0*2"

1 ze
2 d1
3 p1 ph1
  DELTA1 UNBLKGRAD
  p16:gp3
  d16 pl2:f2
  (p3 ph3):f2
  DELTA2
  p16:gp4
  d16
  (p3 ph3):f2
  p16:gp5
  d16

  d20
  (p2 ph1)
  d28
  (p14:sp3 ph1):f2
  4u
```

```
  d29 pl2:f2

  (p3 ph4):f2
  d0
  p2 ph2
  d0
  p16:gp1*EA1
  d16
  (p24:sp7 ph5):f2
  DELTA3
  p16:gp1*EA2
  d16 pl2:f2
  (p3 ph5):f2
  4u BLKGRAD
  go=2 ph31
  d1 mc #0 to 2
    F1EA(calgrad(EA1) & calgrad(EA2), caldel(d0, +in0) & caldel(d20, -in20) & caldel(d28, -in28) &
caldel(d29, +in29) & calph(ph4, +180) & calph(ph31, +180))
exit

ph1=0
ph2=0 0 0 0 2 2 2 2
ph3=0 0 2 2
ph4=0 2
ph5=0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph31=0 2 0 2 0 2 0 2 2 0 2 0 2 0 2 0

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;sp3: f2 channel - shaped pulse (180degree inversion)
;spnam3: Crp60,0.5,20.1
;sp7: f2 channel - shaped pulse (180degree refocussing)
;spnam7: Crp60comp.4
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion
;     = 500usec for Crp60,0.5,20.1
;p16: homospoil/gradient pulse                       [1 msec]
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;d0 : incremented delay (2D)                         [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d16: delay for homospoil/gradient recovery
;d20: decremented delay (J-evolution)
;d28: decremented delay (J-evolution)
;d29: incremented delay (J-evolution)
;cnst6: = 1J(XH)min
;cnst7: = 1J(XH)max
;cnst13: = J(XH) long range
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;in20: = in0
;in28: = in0 * (1 + J(scale) )
;in29: = in0 * J(scale)
;ns: 2 * n
;ds: 16
;td1: number of experiments
;FnMODE: echo-antiecho
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;use gradient ratio:    gp 1 : gp 3 : gp 4 : gp 5
;                         80 :   15 :  -10 :   -5
```

```
;for z-only gradients:
;gpz1: 80%
;gpz3: 15%
;gpz4: -10%
;gpz5: -5%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100

;Processing

;PH_mod(F1): pk
;PHC0(F1): 90

;use xfb and xf2m

;$Id: hmbcetgpjcl2nd,v 1.8 2012/01/31 17:49:23 ber Exp $
```

The following sequences from the Varian (VNMRJ 4.2) pulse program library were used:
5) EXSIDE

```
/* EXSIDE - Gradient Selected phase-sensitive HSQC
                        Adiabatic version

       Features included:
              F1 Axial Displacement

       Paramters:
              sspul :        selects magnetization randomization option
              nullflg:       selects TANGO-Gradient option
              hsglvl:        Homospoil gradient level (DAC units)
              hsgt    :      Homospoil gradient time
              gzlvlE  :      encoding Gradient level
              gtE     :      encoding gradient time
              EDratio :      Encode/Decode ratio
              gstab   :      recovery delay
              j1xh    :      One-bond XH coupling constant

krish krishnamurthy Aug. 95

[REF: V.V. Krishnamurthy, JMR, A121, 33-41 (1996)]

KrishK  -      Revised        : Oct. 2004

*/

#include <standard.h>
#include <chempack.h>

static int      ph4[8]  =  {0,0,0,0,2,2,2,2},
              ph2[2]  =  {0,2},
              ph1[4]  =  {0,0,2,2},
              ph5[8]  =  {0,2,2,0,2,0,0,2};

pulsesequence()

{

  double   selpwrA = getval("selpwrA"),
          selpwA = getval("selpwA"),
          gzlvlA = getval("gzlvlA"),
          gtA = getval("gtA"),
          selpwrB = getval("selpwrB"),
          selpwB = getval("selpwB"),
          gzlvlB = getval("gzlvlB"),
          gtB = getval("gtB"),
                gzlvlE = getval("gzlvlE"),
          gtE = getval("gtE"),
          EDratio = getval("EDratio"),
          gstab = getval("gstab"),
                pwx180 = getval("pwx180"),
                pwxlvl180 = getval("pwxlvl180"),
          hsglvl = getval("hsglvl"),
          hsgt = getval("hsgt"),
          tau,
                evolcorr,
                jdly,
                bird;
  int      icosel,
                ijscale,
          prgcycle = (int)(getval("prgcycle")+0.5),
                phase1=(int)(getval("phase")+0.5);
  char     selshapeA[MAXSTR],
          selshapeB[MAXSTR],
                pwx180ad[MAXSTR],
```

```
                 pwx180adR[MAXSTR];

//synchronize gradients to srate for probetype='nano'
//   Preserve gradient "area"
        gtE = syncGradTime("gtE","gzlvlE",0.5);
        gzlvlE = syncGradLvl("gtE","gzlvlE",0.5);
        gtA = syncGradTime("gtA","gzlvlA",1.0);
        gzlvlA = syncGradLvl("gtA","gzlvlA",1.0);
        gtB = syncGradTime("gtB","gzlvlB",1.0);
        gzlvlB = syncGradLvl("gtB","gzlvlB",1.0);

  getstr("selshapeA",selshapeA);
  getstr("selshapeB",selshapeB);
  getstr("pwx180ad", pwx180ad);
  getstr("pwx180adR", pwx180adR);
  evolcorr=(4*pwx/PI)+2*pw+8.0e-6;
  bird = 1 / (4*(getval("j1xh")));
  tau = 1/(4*(getval("jnxh")));
  ijscale=(int)(getval("jscale") + 0.5);
  jdly=(d2*ijscale);

  icosel = 1;

  assign(ct,v17);
  assign(zero,v18);
  assign(zero,v19);

  if (getflag("prgflg") && (satmode[0] == 'y') && (prgcycle > 1.5))
    {
        hlv(ct,v17);
        mod2(ct,v18); dbl(v18,v18);
        if (prgcycle > 2.5)
           {
               hlv(v17,v17);
               hlv(ct,v19); mod2(v19,v19); dbl(v19,v19);
           }
     }

   settable(t1,4,ph1);
   settable(t2,2,ph2);
   settable(t4,8,ph4);
   settable(t5,8,ph5);

   getelem(t1,v17,v1);
   getelem(t2,v17,v2);
   getelem(t4,v17,v4);
   getelem(t5,v17,oph);

  add(oph,v18,oph);
  add(oph,v19,oph);

  if ((phase1 == 2) || (phase1 == 5))
    icosel = -1;

  initval(2.0*(double)(((int)(d2*getval("sw1")+0.5)%2)),v14);
/*
  mod2(id2, v14);
  dbl(v14,v14);
*/
  add(v2, v14, v2);
  add(oph, v14, oph);

status(A);
   obspower(tpwr);

   delay(5.0e-5);
   if (getflag("sspul"))
```

```
    steadystate();
if (satmode[0] == 'y')
  {
    if ((d1-satdly) > 0.02)
           delay(d1-satdly);
    else
           delay(0.02);
    if (getflag("slpsat"))
      {
           shaped_satpulse("relaxD",satdly,zero);
           if (getflag("prgflg"))
              shaped_purge(v1,zero,v18,v19);
      }
    else
      {
           satpulse(satdly,zero,rof1,rof1);
           if (getflag("prgflg"))
              purge(v1,zero,v18,v19);
      }
  }
else
    delay(d1);

if (getflag("wet"))
  wet4(zero,one);

decpower(pwxlvl);

status(B);

 if (getflag("nullflg"))
 {
   rgpulse(0.5 * pw, zero, rof1, rof1);
   delay(2 * bird);
   decpower(pwxlvl180);
   decshaped_pulse(pwx180ad, pwx180, zero, rof1, rof1);
   rgpulse(2.0 * pw, zero, rof1, rof1);
   delay(2 * bird + 2 * POWER_DELAY);
   decshaped_pulse(pwx180adR, pwx180, zero, rof1, rof1);
   decpower(pwxlvl);
   rgpulse(1.5 * pw, zero, rof1, rof1);
   zgradpulse(hsglvl, hsgt);
   delay(1e-3);
 }

 rgpulse(pw, v1, rof1, rof1);
 delay(tau/2 - gtA - gstab);
 delay(jdly/4);

    zgradpulse(gzlvlA,gtA);
    delay(gstab);
    obspower(selpwrA);
    decpower(pwxlvl180);
    decshaped_pulse(pwx180ad, pwx180, zero, rof1, rof1);
    shaped_pulse(selshapeA,selpwA,zero,rof1,rof1);
    zgradpulse(gzlvlA,gtA);
    delay(gstab);

 delay(tau/2 - gtA - gstab);
 delay(jdly/2);
 delay(tau/2 - gtB - gstab);

    zgradpulse(gzlvlB,gtB);
    delay(gstab);
    obspower(selpwrB);
    decshaped_pulse(pwx180adR, pwx180, zero, rof1, rof1);
    shaped_pulse(selshapeB,selpwB,two,rof1,rof1);
```

```
            obspower(tpwr);
        zgradpulse(gzlvlB,gtB);
        delay(gstab);
            decpower(pwxlvl);

    delay(jdly/4);
    delay(tau/2 - gtB - gstab);

    rgpulse(pw, one, rof1, rof1);
    zgradpulse(hsglvl, 2 * hsgt);
    delay(1e-3);
    decrgpulse(pwx, v2, rof1, 2.0e-6);

      delay(d2/2);
    rgpulse(2 * pw, zero, 2.0e-6, 2.0e-6);
      delay(d2/2);

    delay(gtE + gstab + 2*GRADIENT_DELAY - POWER_DELAY);
    decpower(pwxlvl180);
    decshaped_pulse(pwx180ad, pwx180, zero, rof1, rof1);
    zgradpulse(gzlvlE, gtE);
    delay(gstab + POWER_DELAY + evolcorr);
    decshaped_pulse(pwx180ad, pwx180, zero, rof1, rof1);
    decpower(pwxlvl);

    decrgpulse(pwx, v4, 2.0e-6, rof1);
    zgradpulse(-0.6 * hsglvl, 1.2 * hsgt);
    delay(1e-3);
    rgpulse(pw, zero, rof1, rof1);

    zgradpulse(icosel * 2.0*gzlvlE/EDratio, gtE/2.0);
    delay(gstab);

    delay(tau - gtA - 2*gstab - gtE/2);
        zgradpulse(gzlvlA,gtA);
        delay(gstab);
        obspower(selpwrA);
        decpower(pwxlvl180);
        decshaped_pulse(pwx180ad, pwx180, zero, rof1, rof1);
        shaped_pulse(selshapeA,selpwA,zero,rof1,rof1);
        zgradpulse(gzlvlA,gtA);
        delay(gstab);

    delay(tau - gtA - gstab);

    decshaped_pulse(pwx180adR, pwx180, zero, rof1, rof1);

        decpower(dpwr);
        zgradpulse(gzlvlB,gtB);
        delay(gstab);
        obspower(selpwrB);
        shaped_pulse(selshapeB,selpwB,two,rof1,rof2);
        obspower(tpwr);
        zgradpulse(gzlvlB,gtB);
        delay(gstab);

  status(C);
}
```

The remaining pulse sequences were modified from literature:
6. IPAP Accordion HSQMBC

```
;Pulse Program Code for Bruker:
;On the interference of J(HH) modulation in HSQMBC-IPAP and HMBC-IPAP experiments
;by Josep Saurí and Teodor Parella
;Magn. Reson. Chem. 2013, 51, 509–516, DOI: 10.1002/mrc.3977
;josep_hsqmbcipap
;
;Chad E. Hadden and Daneen T. Angwin
;Magn. Reson. Chem. 2001; 39: 1–8
;
;Define cnst24
;cnst24= 1: HSQMBC
;cnst24= 0: HSQMBC-COSY

;Topspin2.1

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"d0=3u"
"d20=p3"
"d11=30m"
"d12=20u"
"d13=50u"
"d2=1s/(cnst4*2)"
"d14=(1s/(cnst13*4))-d12-50u-p16-d16-p14/2"  ;biggest coupling
"d15=(1s/(cnst14*4))-d12-50u-p16-d16-p14/2"  ;smallest coupling
"in14=(d15-d14)/((td1/2)-1)"  ; IPAP version needs the division by two since it is interleaved
"l0=1"

"in0=inf1/2"

;"DELTA1=d14-p16-50u-4u"
"DELTA=p2+d0*2+50u+p16+d16"
;"DELTA3=d14-50u-p16-d16-50u-p16-d16-p14/2"
;"DELTA4=d14-50u-p16-d16-p14/2"
"DELTA6=50u+p16+d16"
"acqt0=0"

1 ze
2 d1
  4u pl1:f1 pl2:f2
  (p1 ph1)

  50u UNBLKGRAD
  p16:gp5
  d16

  d12 pl0:f2
  d14

  (p1 ph1)
  d2
 (center (p2 ph1) (p14:sp3 ph1):f2 )
  d2
  (p1 ph7)

  d14
  d12 pl2:f2
  50u
  p16:gp5
  d16
```

```
  (p1 ph2)
  4u
  p16:gp3
  d16
  (p3 ph3):f2
  d0
  (p2 ph5)
  d0
  50u
  p16:gp1*EA
  d16 pl0:f2
  (p24:sp7 ph4):f2
  DELTA pl2:f2
  (p3 ph4):f2
  50u
  p16:gp4
  d16

        if "l0 %2 ==1"
        {
        (p1 ph2):f1
        }
        else
        {
        (p1 ph1):f1
        }

  50u
  p16:gp5
  d16

   d12 pl0:f2
   d14

        if "l0 %2 ==1"
        {
        (center (p2 ph1) (p14:sp3 ph1):f2 )
        }
        else
        {
        (p2 ph1)
        }

  d14
  d12 pl2:f2
  50u
  p16:gp5
  d16

if "cnst24==0"
        {
        (p1 ph2):f1
        }
        else
        {

        }

  DELTA6
  p2 ph1
  50u
  p16:gp2
  d16
  4u BLKGRAD

if "l0 %2 ==1"
        {
```

```
            (p3 ph1):f2
            }
            else
            {
            d20
            }

  go=2 ph31
  d1  mc #0 to 2
    F1I(iu0, 2)
    F1EA(igrad EA, id0 & id14 & ip3*2 & ip6*2 & ip31*2)
exit

ph1=0
ph2=1
ph3=0 2
ph4=0
ph5=0
ph6=0
ph7=2
ph22=3
ph24=1
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;sp3: f2 channel - shaped pulse 180 degree (adiabatic)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p16: homospoil/gradient pulse                       [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p28: f1 channel - trim pulse                        [1 msec]
;d0 : incremented delay (2D)                         [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                             [30 msec]
;d13: short delay                                    [4 usec]
;d16: delay for homospoil/gradient recovery
;cnst4: = J(YH)
;cnst13: big long range
;cnst14: small long range
;cnst25: = 0 IP , 1 AP
;cnst24: = 0 HSQMBC-COSY, 1 HSQMBC
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 1 * n
;DS: >= 16
;td1: number of experiments
;FnMODE: echo-antiecho
;use gradient ratio:            gp 1 : gp 2 : gp 3 : gp 4 : gp 5
;                                 80 : 20.1 : 33 : 50 : 17 for C-13
;                                 80 :  8.1 : 33 : 50 : 17 for N-15

;for z-only gradients:
;gpz1: 80%
;gpz2: 20.1%
;gpz3: 33%
;gpz4: 50%
;gpz5: 17%

;use gradient files:
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.100
```

7. IPAP refocussed HSQMBC

```
;Pulse Program Code for Bruker:
;On the interference of J(HH) modulation in HSQMBC-IPAP and HMBC-IPAP experiments
;by Josep Saurí and Teodor Parella
;Magn. Reson. Chem. 2013, 51, 509–516, DOI: 10.1002/mrc.3977
;josep_hsqmbcipap
;Define cnst24
;cnst24= 1: HSQMBC
;cnst24= 0: HSQMBC-COSY

;Topspin2.1

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p4=p3*2"
"d0=3u"
"d20=p3"
"d11=30m"
"d13=50u"
"d16=200u"
"d2=1s/(cnst4*2)"
"d14=1s/(cnst13*4)"
"in0=inf1/2"
"d20=p3"
"d22=p1"
"l0=1"

"DELTA1=d14-p16-50u-4u"
"DELTA=p2+d0*2+50u+p16+d16"
"DELTA3=d14-50u-p16-d16-50u-p16-d16-p14/2"
"DELTA4=d14-50u-p16-d16-p14/2"
"DELTA6=50u+p16+d16"

"acqt0=0"
baseopt_echo

1 ze
2 d1
  4u pl1:f1 pl0:f2
  (p1 ph1)

  50u UNBLKGRAD
  p16:gp5
  d16
  DELTA4

  (p1 ph1)
  d2
 (center (p2 ph1) (p14:sp3 ph1):f2 )
  d2
  (p1 ph7)

  DELTA4
  50u
  p16:gp5
  d16
  (p1 ph2)
  3u pl2:f2
  p16:gp3
  d16
  (p3 ph3):f2
  d0
  (p2 ph5)
```

```
 d0
 50u pl0:f2
 p16:gp1*EA
 d16
 (p24:sp7 ph4):f2
 DELTA pl2:f2
 (p3 ph4):f2
 50u
 p16:gp4
 d16

        if "l0 %2 ==1"
        {
        (p1 ph2):f1
        }
        else
        {
        (p1 ph1):f1
        }

 50u
 p16:gp5
 d16
 DELTA4 pl0:f2

        if "l0 %2 ==1"
        {
        (center (p2 ph1) (p14:sp3 ph1):f2 )
        }
        else
        {
        (p2 ph1)
        }

 DELTA4
 50u
 p16:gp5
 d16

if "cnst24==0"
        {
        (p1 ph2):f1
        }
        else
        {
        d22
        }


 DELTA6 pl2:f2
 p2 ph1
 50u
 p16:gp2
 d16

 if "l0 %2 ==1"
        {
        (p3 ph1):f2
        }
        else
        {
        d20
        }

 4u BLKGRAD

 go=2 ph31
```

```
  d1  mc #0 to 2
     F1I(iu0, 2)
     F1EA(igrad EA, id0 & ip3*2 & ip6*2 & ip31*2)
exit

ph1=0
ph2=1
ph3=0 2
ph4=0
ph5=0
ph6=0
ph7=2
ph22=3
ph24=1
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;sp3: f2 channel - shaped pulse 180 degree (adiabatic)
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion (adiabatic)
;p16: homospoil/gradient pulse                        [1 msec]
;p21: f3 channel -  90 degree high power pulse
;p22: f3 channel - 180 degree high power pulse
;p28: f1 channel - trim pulse                         [1 msec]
;d0 : incremented delay (2D)                          [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                              [30 msec]
;d13: short delay                                     [4 usec]
;d16: delay for homospoil/gradient recovery
;d26: 1/(4J(XH)) delay in G-BIRD
;cnst4: = 1J(XH) one J-coupling for G-BIRD supression
;cnst13: = nJ(XH) long range coupling
;cnst25: = 0 IP , 1 AP
;cnst24: = 0 HSQMBC-COSY, 1 HSQMBC
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 1 * n
;DS: >= 16
;td1: number of experiments
;FnMODE: echo-antiecho

;for z-only gradients:
;gpz1: 80%
;gpz2: 20.1%
;gpz3: 33%
;gpz4: 50%
;gpz5: 17%

;use gradient files:
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.100
```

8. IPAP PIP-HSQMBC

```
;piphsqmbc
;Pure In-Phase HSQMBC  with final adiabatic z-filter and perfect echo gradient
;optional IPAP using const25
;"Pure In-Phase heteronuclear correlation NMR experiments" by
;L. Castañar, J. Saurí, R. T. Williamson, A. Virgili and T. Parella
;Angew. Chem. Int. Ed. 2014, 53, 8379 –8382, DOI: 10.1002/anie.201404136

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"p5=p1*0.5"
"d6=1s/(cnst2*4)"
"d2=1s/(cnst16*2)-(p14/2)"
"d4=1s/(cnst17*2)-(p14/2)"
"d0=3u"
"d11=30m"
"in0=inf1/2"
"l0=1"

"DELTA2=d6-p16-d16-larger(p2,p14)/2"
"DELTA=p16+d16+p2+d0*2"
"DELTA4=p16+d16"
"acqt0=0"
baseopt_echo

1 ze
2 d1 pl1:f1 pl0:f2
3 d11 UNBLKGRAD
  (p5 ph1):f1
  d4
  (center (p2 ph2) (p14:sp3 ph6):f2)
  d4 pl2:f2
  (p5 ph7):f1 (p3 ph6):f2

  DELTA2 pl0:f2
  p16:gp3
  d16
  (p1 ph2)
  d2
  (center (p2 ph2) (p14:sp3 ph6):f2 )
  d2
  (p1 ph8)
  p16:gp3
  d16
  DELTA2 pl2:f2

  (p1 ph2) (p3 ph3):f2
  d0
  (p2 ph5)
  d0
  p16:gp1*EA
  d16 pl0:f2
  (p24:sp7 ph4):f2
  DELTA pl2:f2
  (ralign (p1 ph1) (p3 ph4):f2 )

if "l0 %2 == 1"
        {
        DELTA2 pl0:f2
        (center (p2 ph1) (p14:sp3 ph1):f2 )
        DELTA2 pl2:f2
        }
        else
```

```
            {
              d6
             (p2 ph2)
             d6
            }

  if "l0 %2 == 1"
            {
            (p1 ph2):f1
            }
            else
            {
            (p1 ph1):f1
            }

  d12 pl0:f1
  300u gron0
  (p32:sp29 ph1):f1
  300u groff
  d12 pl1:f1
  p1 ph1

  DELTA4
  p2 ph1
  DELTA4
  p1 ph2
  DELTA4
  p2 ph1
  p16:gp2
  d16 BLKGRAD

  go=2 ph31
  d1 mc #0 to 2
    F1I(iu0, 2)
    F1EA(igrad EA, id0 & ip3*2 & ip6*2 & ip31*2)
exit

ph1=0
ph2=1
ph3=0 2
ph4=0
ph5=0
ph6=0
ph7=2
ph8=3
ph31=0 2

;pl0 : 0W
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;sp3: f2 channel - shaped pulse 180 degree
;sp7: f2 channel - shaped pulse 180 degree
;sp29: f2 channel - shaped adiabatic pulse 180 degree
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p4 : f2 channel - 180 degree high power pulse
;p5 : f1 channel - 90 degree high power pulse (TANGO)
;p14: f2 channel - 180 degree shaped pulse for inversion
;p24: f2 channel - 180 degree shaped pulse for refocusing
;p32: f2 channel - 180 degree shaped pulse for adiabatic z-filter
;p16: homospoil/gradient pulse
;d0 : incremented delay (2D)                    [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d2 : 1/(2*J(HX) delay in BIRD suppression
;d4 : 1/(2*J(HX) delay in TANGO suppression
;cnst25:0 (IP) 1 (AP)
```

```
;cnst2: ~8 Hz (should be optimized to a value between 3-12 Hz)
;cnst16: ~120 J(HX) for BIRD suppression
;cnst17: ~145 J(HX) for TANGO suppression (cnst16 != cnst17)
;d6 : 1/(4J)XH (long range coupling constant)
;d16: delay for homospoil/gradient recovery
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 2 * n
;DS: >= 2
;td1: number of experiments
;FnMODE: echo-antiecho

;use gradient ratio:           gp 1 : gp 2 : gp 3
;                                80 : 20.1 : 17 for C-13
;                                80 :  8.1 : 17 for N-15

;for z-only gradients:
;gpz0: 3%
;gpz1: 80%
;gpz2: 20.1% for C-13, 8.1% for N-15
;gpz3: 17%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
```

9. IPAP HMBC

```
;Pulse Program Code for Bruker:
;On the interference of J(HH) modulation in HSQMBC-IPAP and HMBC-IPAP experiments
;by Josep Saurí and Teodor Parella
;Magn. Reson. Chem. 2013, 51, 509–516, DOI: 10.1002/mrc.3977
;josep_hmbcipap
;Define cnst24
;cnst24= 1: HMBC
;cnst24= 0: HMBC-COSY

;Topspin2.1

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"d0=3u"
"d20=p3"
"d6=1s/(cnst13*2)"
"d20=p3"
"d22=p1"
"DELTA5=d6/2"
"DELTA1=1s/(2 * cnst6)-p16-d16"
"DELTA2=1s/(2 * cnst7)-p16-d16"
"DELTA3=d6-p16-d16-4u"
"DELTA4=p2+d0*2"
"DELTA=p16+d16+50u"
"l0=1"
"in0=inf1/2"

"acqt0=0"
baseopt_echo

1 ze
2 d1 pl1:f1 pl2:f2
if "l0 %2 ==1"
        {
        p1 ph6
        }
        else
        {
        p1 ph1
        }

 DELTA1 UNBLKGRAD
 p16:gp3
 d16
 (p3 ph3):f2
 DELTA2
 p16:gp4
 d16
 (p3 ph3):f2
 4u
 p16:gp5
 d16
 DELTA3
 (p3 ph4):f2
 d0
 p2 ph2
 d0
 p16:gp1
 d16 pl0:f2
 (p24:sp7 ph5):f2
 DELTA4 pl2:f2
 p16:gp2
```

```
  d16
  (p3 ph5):f2

  DELTA5 pl0:f2

if "l0 %2 ==1"
         {
         (center (p2 ph1) (p14:sp3 ph1):f2 )
         }
         else
         {
         (p2 ph1)
  }

DELTA5 pl2:f2

if "cnst24==0"
         {
         p1 ph6
         }
         else
         {
         d22
  }

  DELTA
  (p2 ph1)
  50u
  p16:gp6*EA
  d16
  4u BLKGRAD

  if "l0 %2 ==1"
         {
         (p3 ph1):f2
         }
         else
         {
         d20
  }

  go=2 ph31
  d1 mc #0 to 2
    F1I(iu0, 2)
    F1EA(igrad EA, id0 & ip4*2 & ip31*2)
exit


ph1=0
ph2=0
ph3=0
ph4=0 2
ph5=0
ph6=1
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;sp7: f2 channel - shaped pulse (180degree refocussing)
;spnam7: Crp60comp.4
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p16: homospoil/gradient pulse                     [1 msec]
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;d0 : incremented delay (2D)                       [3 usec]
```

```
;d1 : relaxation delay; 1-5 * T1
;d6 : delay for evolution of long range couplings (1/2Jlr)
;d16: delay for homospoil/gradient recovery
;cnst6: = 1J(XH)min
;cnst7: = 1J(XH)max
;cnst13: = J(XH) long range
;cnst24: = 1 HMBC 0 HMBC-COSY
;cnst25: = 0 IP 1 AP
;in0: 1/(2 * SW(X)) = DW(X)
;nd0: 2
;NS: 2 * n
;DS: 16
;td1: number of experiments
;FnMODE: echo-antiecho

;use gradient ratio:       40 : -40 :  15 :  -10 :   -5 :  20.1 for C-13
;                                                         21.4 for F-19
;                                                          8.1 for N-15

;for z-only gradients:
;gpz1: 40%
;gpz2: -40%
;gpz3: 15%
;gpz4: -10%
;gpz5: -5%
;gpz6:  20.1%

;use gradient files:
;gpnam1: SINE.100
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.100
;gpnam6: SINE.100

;Processing
;PH_mod(F1): pk (or no)
;use xfb and xf2m
;$Id: hmbcetgpl2nd,v 1.2.8.1 2005/11/1
```

10. HOBS-HSQMBC-1

;Pulse Program Code for Bruker:

```
;pp:hobs_hsqmbc
;2D Experiment
;HOmodecoupled Band-Selective HSQMBC experiment (HOBS-HSQMBC)
;Instant homonuclear broadband decoupling is used during the acquisition
;Long-range 1H-13C correlation
;Phase sensitive using Echo/Antiecho-TPPI gradient selection

;L. Castañar, J. Sauri, P. Nolis, A. Virgili & T. Parella,
;Journal of Magnetic Resonance 238 (2014) 63-69

;Avance III version (07/07/2013)
;Topspin2.1&Topspin3.1

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>
#include <De.incl>

;#define hobs

"p2=p1*2"
"d4=1s/(cnst2*4)"
"d11=30m"
"d0=3u"
"in0=inf1/2"
"d27=p3"
"d12=20u"
"d16=200u"
"DELTA1=50u+p16+d16"
"DELTA2=d4-larger(p12,p14)/2-50u-p16-d16"
"DELTA3=d4-larger(p12,p14)/2-50u-p16-d16-50u-p16-d16"
"DELTA=d12+p16+d16+50u+p2+d0*2"
"acqt0=0"
#ifdef hobs
dwellmode explicit
"d2=aq/l0"
"d3=d2/2"
"l2=l0-1"
#else
#endif

1 ze
2 d1
  d12

3 (p1 ph1)
  50u UNBLKGRAD
  p16:gp6
  d16
  DELTA2 pl0:f2 pl0:f1
  (center (p12:sp2 ph1) (p14:sp3 ph6):f2 )
  DELTA2 pl2:f2 pl1:f1
  50u
  p16:gp6
  d16
  (p1 ph2)
  50u
  p16:gp3
  d16
  (p3 ph3):f2

  d0
  (p2 ph5)
  d0
```

```
 50u
 p16:gp1*EA
 d16 pl0:f2
 d12
 (p24:sp7 ph4):f2
 DELTA pl2:f2

 (p3 ph4):f2
 50u
 p16:gp4
 d16

 if "cnst25==0"
            {
            (p1 ph2):f1
            }
            else
            {
            (p1 ph1):f1
 }

 50u
 p16:gp5
 d16
 DELTA2 pl0:f2 pl0:f1

 if "cnst25==0"
            {
            (center (p12:sp2 ph1) (p14:sp3 ph1):f2 )
            }
            else
            {
            (p12:sp2 ph1)
 }

 DELTA3 pl2:f2 pl1:f1
 50u
 p16:gp5
 d16
 50u
 p16:gp2
 d16

 if "cnst25==0"
            {
            (p3 ph1):f2
            }
            else
            {
            d27
 }

#ifdef hobs

  ACQ_START(ph30,ph31)

  0.05u DWL_CLK_ON
  0.1u REC_UNBLK
  d3
  0.05u DWL_CLK_OFF
  0.1u REC_BLK

  p19:gp8
  d16 pl1:f1
  (p2 ph8):f1
  p19:gp8
```

```
   d16

   p19:gp7
   d16 pl0:f1
   (p12:sp2 ph7):f1
   p19:gp7
   d16

5

   0.05u DWL_CLK_ON
   0.1u REC_UNBLK
   d2:r
   0.05u DWL_CLK_OFF
   0.1u REC_BLK

   p19:gp8
   d16 pl1:f1
   (p2 ph8):f1
   p19:gp8
   d16

   p19:gp7
   d16 pl0:f1
   (p12:sp2 ph7):f1
   p19:gp7
   d16

lo to 5 times l2

   0.05u DWL_CLK_ON
   0.1u REC_UNBLK
   d3
   15m
   0.05u DWL_CLK_OFF
   0.1u REC_BLK
   d12 BLKGRAD

10 rcyc=2

#else
go=2 ph31
#endif

  d1 mc #0 to 2
    F1EA(igrad EA, id0 & ip3*2 & ip6*2 & ip31*2)
exit

ph1=0
ph2=1
ph3=0 2
ph4=0
ph5=0
ph6=0
ph7=0
ph8=0
ph30=0
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  90 degree high power pulse
;p12 : f1 channel - 180 degree band-selective pulse [ms]
;p16: homospoil/gradient pulse [1 ms]
;p14: f2 channel - 180 degree shaped pulse for inversion
;p19: homospoil/gradient pulse [500 us]
```

```
;p24: f2 channel - 180 degree shaped pulse for refocussing
;sp2: f1 channel - shaped pulse power level for 180 degree  band-selective excitation
;spnam2: shaped pulse for 180 degree selective excitation [REBURP]
;d0 : incremented delay (2D)
;d1 : relaxation delay; 1-5 * T1
;d4 : 1/(4*nJ(XH))
;d16: delay for homospoil/gradient recovery
;cnst2: = nJ(XH)
;NS: 2 * n
;DS: 16
;FnMODE: echo-antiecho
;l2: number of concatenated loops
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;cnst25: =0 (IP)
;cnst25: =1 (AP)

;use gradient ratio:          gp 1 : gp 2 : gp 3 : gp 4 : gp 5 : gp 6 : gp 7 : gp 8
;                            80 : 20.1 :  33  :  50  :  11  : 17   :  63  : 41

;for z-only gradients:
;gpz1: 80%
;gpz2: 20.1% for C-13, 8.1% for N-15
;gpz3: 33%
;gpz4: 50%
;gpz5: 11%
;gpz6: 17%
;gpz7: 63%
;gpz8: 41%

;use gradient files:
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.100
;gpnam6: SINE.100
;gpnam7: SINE.100
;gpnam8: SINE.100
```

11. HOBS-HSQMBC-2

```
;Pulse Program Code for Bruker:

;pp:behobs_hsqmbc
;2D Experiment
;Broadband-hEterodecoupled and HOmodecoupled Band-Selective HSQMBC experiment (BEHOBS-
HSQMBC)
;Instant homonuclear broadband decoupling is used during the acquisition
;Long-range 1H-13C correlation
;phase sensitive using Echo/Antiecho-TPPI gradient selection

;L. Castañar, J. Sauri, P. Nolis, A. Virgili & T. Parella,
;Journal of Magnetic Resonance 238 (2014) 63-69

;Avance III version (07/07/2013)
;Topspin2.1&Topspin3.1


#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>
#include <De.incl>



;#define hobs

"p2=p1*2"
"d4=1s/(cnst2*4)"
"d11=30m"
"d0=3u"
"in0=inf1/2"
"d12=12u"
"d16=200u"
"DELTA1=50u+p16+d16"
"DELTA2=d4-larger(p12,p14)/2-50u-p16-d16"
"DELTA3=d4-larger(p12,p14)/2-50u-p16-d16-50u-p16-d16"
"DELTA=p2+d0*2"
"acqt0=0"

#ifdef hobs
dwellmode explicit
"d2=aq/l0"
"d3=d2/2"
"l2=l0-1"
#else
#endif


1 ze
2 d1
  d12 pl1:f1

3 (p1 ph1)
  50u UNBLKGRAD
  p16:gp6
  d16
  DELTA2 pl0:f2 pl0:f1
  (center (p12:sp2 ph1) (p14:sp3 ph6):f2 )
  DELTA2 pl2:f2 pl1:f1
  50u
  p16:gp6
  d16
  (p1 ph2)
  50u
  p16:gp3
  d16
```

```
(p3 ph3):f2

d0
(p2 ph5)
d0

50u
p16:gp1*EA
d16 pl0:f2
d12
(p24:sp7 ph4):f2
50u
p16:gp1*EA*-1
d16
d12
DELTA pl2:f2

(p3 ph4):f2
50u
p16:gp4
d16

if "cnst25==0"
      {
      (p1 ph2):f1
      }
      else
      {
      (p1 ph1):f1
}

50u
p16:gp5
d16
DELTA2 pl0:f2 pl0:f1

if "cnst25==0"
      {
      (center (p12:sp2 ph1) (p14:sp3 ph1):f2 )
      }
      else
      {
      (p12:sp2 ph1)
}

DELTA3 pl2:f2 pl1:f1
50u
p16:gp5
d16
50u
p16:gp2
d16

if "cnst25==0"
      {
      (p3 ph1):f2
      }
      else
      {
      17.5u
}

d16


#ifdef hobs
```

```
   ACQ_START(ph30,ph31)

   0.05u DWL_CLK_ON
   0.1u REC_UNBLK
   d3
   0.05u DWL_CLK_OFF
   0.1u REC_BLK


   p19:gp8
   d16 pl1:f1
   (p2 ph8):f1
   p19:gp8
   d16

   p19:gp7
   d16 pl0:f1
   (p12:sp2 ph7):f1
   p19:gp7
   d16

5  0.05u DWL_CLK_ON
   0.1u REC_UNBLK
   d2:r
   0.05u DWL_CLK_OFF
   0.1u REC_BLK


   p19:gp8
   d16 pl1:f1
   (p2 ph8):f1
   p19:gp8
   d16

   p19:gp7
   d16 pl0:f1
   (p12:sp2 ph7):f1
   p19:gp7
   d16

lo to 5 times l2

   0.05u DWL_CLK_ON
   0.1u REC_UNBLK
   d3
   15m
   0.05u DWL_CLK_OFF
   0.1u REC_BLK
   d12 BLKGRAD

10 rcyc=2

#else
go=2 ph31
#endif

  d1 mc #0 to 2
     F1EA(igrad EA, id0 & ip3*2 & ip6*2 & ip31*2)
exit


ph1=0
ph2=1
ph3=0 2
ph4=0
ph5=0
ph6=0
```

```
ph7=0
ph8=0
ph30=0
ph31=0 2

;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;p1 : f1 channel -  high power pulse
;p12 : f1 channel - 180 degree band-selective pulse [ms]
;p14: f2 channel - 180 degree shaped pulse for inversion
;p16: homospoil/gradient pulse [1 ms]
;p19: homospoil/gradient pulse [500 us]
;p24: f2 channel - 180 degree band-selective pulse [ms]
;sp2: f1 channel - shaped pulse power level for 180 degree  band-selective excitation
;sp7: f2 channel - shaped pulse power level for 180 degree  band-selective excitation
;spnam2: f1 channel - shaped pulse for 180 degree selective excitation [REBURP]
;spnam7: f2 channel - shaped pulse for 180 degree selective excitation [REBURP]
;d0 : incremented delay (2D)
;d1 : relaxation delay; 1-5 * T1
;d4 : 1/(4*nJ(XH))
;d16: delay for homospoil/gradient recovery
;cnst2: = nJ(XH)
;NS: 2 * n
;DS: 16
;FnMODE: echo-antiecho
;l2: number of concatenated loops
;cpd2: decoupling according to sequence defined by cpdprg2 [garp]
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
;cnst25: =0 (IP)
;cnst25: =1 (AP)


;use gradient ratio:            gp 1 : gp 2 : gp 3 : gp 4 : gp 5 : gp 6 : gp 7 : gp 8
;                               40 : 20.1 :  33  :  50  :  11  : 17   :  63  : 41


;for z-only gradients:
;gpz1: 40%
;gpz2: 20.1% for C-13, 8.1% for N-15
;gpz3: 33%
;gpz4: 50%
;gpz5: 11%
;gpz6: 17%
;gpz7: 63%
;gpz8: 41%

;use gradient files:
;gpnam1: SINE.100
;gpnam2: SINE.100
;gpnam3: SINE.100
;gpnam4: SINE.100
;gpnam5: SINE.100
;gpnam6: SINE.100
;gpnam7: SINE.100
;gpnam8: SINE.100
```

[1] Marquez, B. L.; Gerwick, W. H.; Thomas Williamson, R. *Magn. Reson. Chem.* **2001**, *39* (9), 499.
[2] Edden, R. A. E.; Keeler, J. *J. Magn. Reson.* **2004**, *166* (1), 53.
[3] Butts, C. P.; Heise, B.; Tatolo, G. *Org. Lett.* **2012**, *14* (13), 3256.